** Preamble **

clear all
set more off

local path_input  "C:/Users/$user/Dropbox/$project/master/output/data"
local path_output "C:/Users/$user/Dropbox/$project/master/output/tables"

* Note: local directory is the output folder 
cd "`path_output'"

			** Loading the input file **

use `path_input'/DataExperiment15Abril, clear


forvalues i=1/4{
	foreach var in SocialismoComunismo LutaarmadaeInvasãodeterras SistemaBancário divida Imprensa QuestõesSociais Inflação Sarney PrevidênciaSocial Alagoas CongressoNacional Constituiçãode1988 Habitação Greves Educação Orçamentopúblico ServiçosPúblicos CampanhaEleitoral RelaçãoPatrõesxTrabalhadores Oposiçãodocandidatoderrotado Corrupção{

		tostring `var'`i'                     , replace
		replace  `var'`i'=""  if `var'`i'=="."
		replace  `var'`i'="1" if `var'`i'!=""
		destring `var'`i'                     , replace
		replace  `var'`i'=0   if `var'`i'==.
		
	}
}

forvalues i=1/4{

	rename SocialismoComunismo`i' socialismocomunismo`i'
	rename divida`i'              DívidaPública`i'
	rename Imprensa`i'            imprensa`i'
	rename Sarney`i'              sarney`i'
	rename Alagoas`i'             alagoas`i'
	rename CongressoNacional`i'   congressonacional`i'
	rename Greves`i'              greves`i'
	rename CampanhaEleitoral`i'   campanhaeleitoral`i'

}

foreach var in socialismocomunismo LutaarmadaeInvasãodeterras SistemaBancário DívidaPública imprensa QuestõesSociais Inflação sarney PrevidênciaSocial alagoas congressonacional Constituiçãode1988 Habitação greves Educação Orçamentopúblico ServiçosPúblicos campanhaeleitoral RelaçãoPatrõesxTrabalhadores Oposiçãodocandidatoderrotado Corrupção{

	gen `var'=`var'1+`var'2+`var'3+`var'4

}


merge m:1 trecho using `path_input'/TemasTrechos
keep if _m!=2

foreach var in socialismocomunismo LutaarmadaeInvasãodeterras SistemaBancário DívidaPública imprensa nordeste Inflação sarney PrevidênciaSocial alagoas congressonacional Constituiçãode1988 Habitação greves Educação ServiçosPúblicos Orçamentopúblico campanhaeleitoral RelaçãoPatrõesxTrabalhadores Oposiçãodocandidatoderrotado Corrupção{

	cap rename G_`var' g_`var'

}

gen aux_topicos=0
foreach var in socialismocomunismo LutaarmadaeInvasãodeterras SistemaBancário DívidaPública imprensa Inflação sarney PrevidênciaSocial alagoas congressonacional Constituiçãode1988 Habitação greves Educação ServiçosPúblicos Orçamentopúblico campanhaeleitoral RelaçãoPatrõesxTrabalhadores Oposiçãodocandidatoderrotado Corrupção{

	replace aux_topicos=aux_topicos+1 if `var'==1 & g_`var'==1

}

replace aux_topicos=. if trecho=="CF"
replace aux_topicos=. if trecho=="JN_S_VP"
replace aux_topicos=. if trecho=="JN_VP"

replace  validacao_pergunta="3"  if validacao_pergunta=="Quanto &eacute; sete menos&nbsp;quatro?"
replace  validacao_pergunta="72" if validacao_pergunta=="Quanto &eacute; oitenta e quatro menos&nbsp;doze?"
replace  validacao_pergunta="12" if validacao_pergunta=="Quanto &eacute; vinte menos oito?"
replace  validacao_pergunta="9"  if validacao_pergunta=="Quanto é doze menos&nbsp;três?"
destring validacao_pergunta, replace

replace  validacao_resposta="12" if validacao_resposta=="DOZE"
replace  validacao_resposta="9"  if validacao_resposta=="Nove."
replace  validacao_resposta="72" if validacao_resposta=="Setenta e dois."
replace  validacao_resposta="3"  if validacao_resposta=="Três"
replace  validacao_resposta="12" if validacao_resposta=="doze"
replace  validacao_resposta="9"  if validacao_resposta=="nove"
replace  validacao_resposta="12" if validacao_resposta=="setenta e dois"
replace  validacao_resposta="3"  if validacao_resposta=="três"
destring validacao_resposta, replace

gen aux_validacao_mat=validacao_pergunta!=validacao_resposta

gen hours=hours(date_modified-date_created)
gen minutes=hours*60

*dropando observações que não são válidas
gen 	dropable=1 if minutes<5
replace dropable=1 if minutes>180
replace dropable=1 if aux_validacao_mat==1


reg vota_lula   treatment_JN if dropable!=1
reg ganhou_lula treatment_JN if dropable!=1
reg vota_lula   treatment_JN if dropable==1
reg ganhou_lula treatment_JN if dropable==1

gen control_JN=.
replace control_JN=1 if trecho      =="P1_B1"
replace control_JN=1 if trecho      =="P1_B3"
replace control_JN=1 if trecho      =="P1_B2"
replace control_JN=1 if trecho      =="P4_B3"
replace control_JN=1 if trecho      =="P2_B4"
replace control_JN=1 if trecho      =="P3_B4"
replace control_JN=1 if treatment_JN==1


replace escolaridade="Ensino médio completo" if escolaridade=="Menos do que o ensino médio"

gen     regiao="N"  if estado=="Amapá"
replace regiao="N"  if estado=="Amazonas"
replace regiao="NE" if estado=="Bahia"
replace regiao="NE" if estado=="Ceara"
replace regiao="CO" if estado=="Distrito Federal" 
replace regiao="SD" if estado=="Minas Gerais" 
replace regiao="N"  if estado=="Para" 
replace regiao="NE" if estado=="Paraiba" 
replace regiao="S"  if estado=="Parana" 
replace regiao="NE" if estado=="Pernambuco" 
replace regiao="NE" if estado=="Piaui" 
replace regiao="NE" if estado=="Rio Grande do Norte" 
replace regiao="S"  if estado=="Rio Grande do Sul" 
replace regiao="SD" if estado=="Rio de Janeiro" 
replace regiao="N"  if estado=="Rondônia"
replace regiao="S"  if estado=="Santa Catarina"
replace regiao="SD" if estado=="São Paulo"



tab regiao
gen CENTRO_OESTE = (regiao=="CO")
gen SUDESTE      = (regiao=="SD")
gen NORTE        = (regiao=="N")
gen NORDESTE     = (regiao=="NE")
gen SUL          = (regiao=="S")

tab escolaridade
gen undergrad   = (escolaridade=="Bacharelado/Licenciatura" | escolaridade=="Ensino superior incompleto")
gen high_school = (escolaridade=="Diploma técnico" | escolaridade=="Ensino médio completo")
gen grad        = (escolaridade=="Pós-graduação")
drop if dropable==1

								
								** Table A1: Balance check, experimental sample **

local electoral       "vota_lula ganhou_lula"
local characteristics "minutes idade posicao_politica undergrad high_school grad SUL SUDESTE CENTRO_OESTE NORDESTE NORTE"

    
label var vota_lula        "Vote for Lula Today"
label var ganhou_lula      "Lula won Debate"

label var minutes          "Survey time (min.)"
label var idade 		   "Age"	
label var posicao_politica "Left to Right Scale"
label var undergrad 	   "Undergraduate"
label var high_school 	   "High-school" 
label var grad 			   "Graduate"
label var SUL 			   "South"
label var SUDESTE  		   "Southeast"	
label var CENTRO_OESTE 	   "Midwest"	
label var NORDESTE 		   "Northeast"	
label var NORTE			   "North"


foreach v of varlist * {

	label variable `v' `"\hspace{0.1cm} `: variable label `v''"'
	
}

										** Producing table ** 
  
											** Panel A **


balancetable (mean if treatment_JN==1) (mean if treatment_JN==0) (diff treatment_JN) `electoral'      using Table_balance_experiment_panel_A.tex,  replace /// 
			  wide(mean1 sd1 mean2 sd2 pval3) 															  									  		  	   /// 
			  pvalues varlabel nonumbers noobservations nopar          														  				  		  	   ///
			  ctitles("Mean" "Std. dev." "Mean" "Std. dev." "p-value") 														  				  		  	   ///
			  groups("Treatment" "Control", pattern(1 0 1 0 1) prefix(\multicolumn{@span}{c}{) suffix(}))						

	
			  
											** Panel B **

balancetable (mean if treatment_JN==1) (mean if treatment_JN==0) (diff treatment_JN) `characteristics' using Table_balance_experiment_panel_B.tex,  replace /// 
			  wide(mean1 sd1 mean2 sd2 pval3) 															  									  				/// 
			  pvalues varlabel nonumbers noobservations nopar          														  				  				///
			  ctitles("Mean" "Std. dev." "Mean" "Std. dev." "p-value") 														  				  				///
			  groups("Treatment" "Control", pattern(1 0 1 0 1) prefix(\multicolumn{@span}{c}{) suffix(}))													///
			  prefoot("Observations & 54 & & 57 & & 111 \\")
			
			 
include "https://raw.githubusercontent.com/steveofconnell/PanelCombine/master/PanelCombine.do"  

panelcombine, use(Table_balance_experiment_panel_A.tex Table_balance_experiment_panel_B.tex) 	///
paneltitles("Perception about Candidates" "Participant Characteristics") columncount(6)    		/// 
save(Online_Experiment.tex) cleanup










